Problem:
Score : 600 points

Problem Statement

Given are an integer N and arrays S, T, U, and V, each of length N.
Construct an N×N matrix a that satisfy the following conditions:

a_{i,j} is an integer.

0 \leq a_{i,j} \lt 2^{64}.

If S_{i} = 0, the bitwise AND of the elements in the i-th row is U_{i}.

If S_{i} = 1, the bitwise OR of the elements in the i-th row is U_{i}.

If T_{i} = 0, the bitwise AND of the elements in the i-th column is V_{i}.

If T_{i} = 1, the bitwise OR of the elements in the i-th column is V_{i}.

However, there may be cases where no matrix satisfies the conditions.

Constraints

All values in input are integers.

1 \leq N \leq 500

0 \leq S_{i} \leq 1

0 \leq T_{i} \leq 1

0 \leq U_{i} \lt 2^{64}

0 \leq V_{i} \lt 2^{64}

Input

Input is given from Standard Input in the following format:

N
S_{1} S_{2} ... S_{N}
T_{1} T_{2} ... T_{N}
U_{1} U_{2} ... U_{N}
V_{1} V_{2} ... V_{N}

Output

If there exists a matrix that satisfies the conditions, print one such matrix in the following format:

a_{1,1} ... a_{1,N}
:
a_{N,1} ... a_{N,N}

Note that any matrix satisfying the conditions is accepted.

If no matrix satisfies the conditions, print -1.

Sample Input 1

2
0 1
1 0
1 1
1 0

Sample Output 1

1 1
1 0

In Sample Input 1, we need to find a matrix such that:

the bitwise AND of the elements in the 1-st row is 1;

the bitwise OR of the elements in the 2-nd row is 1;

the bitwise OR of the elements in the 1-st column is 1;

the bitwise AND of the elements in the 2-nd column is 0.

Sample Input 2

2
1 1
1 0
15 15
15 11

Sample Output 2

15 11
15 11

Solution in Fortran (GNU Fortran 9.2.1) (judge verdict: Accepted)

program i_hate_matrix_construction
  implicit none
  integer, parameter :: intkind = 16
  integer :: n, i, j, k
  integer, dimension(500) :: s = 0, t = 0, p = 0, q = 0, x = 0, y = 0
  integer(intkind), dimension(500) :: u = 0, v = 0
  integer(intkind), dimension(500, 500) :: a = 0
  read(*,*) n
  read(*,*) s(1:n)
  read(*,*) t(1:n)
  read(*,*) u(1:n)
  read(*,*) v(1:n)
  do k = 0, 64
    do i = 1, n
      p(i) = pattern(s(i), btest(u(i), k))
      q(i) = pattern(t(i), btest(v(i), k))
    end do
    x(1:n) = 0
    y(1:n) = 0
    do i = 1, n
      do j = 1, n
        if (p(i) + q(j) == 1) call halt()
        if (p(i) == 1 .or. q(j) == 1 .or. p(i) + q(j) == 6) then
          a(i, j) = ibset(a(i, j), k)
          x(i) = x(i) + 1
          y(j) = y(j) + 1
        end if
      end do
    end do
    j = 1
    do i = 1, n
      if (p(i) == 2 .and. x(i) == n) call halt()
      if (.not.(p(i) == 3 .and. x(i) == 0)) cycle
      do while (j <= n .and. .not.(q(j) == 2 .and. y(j) < n - 1))
        j = j + 1
      end do
      if (j > n) call halt()
      a(i, j) = ibset(a(i, j), k)
      x(i) = x(i) + 1
      y(j) = y(j) + 1
    end do
    i = 1
    do j = 1, n
      if (q(j) == 2 .and. y(j) == n) call halt()
      if (.not.(q(j) == 3 .and. y(j) == 0)) cycle
      do while (i <= n .and. .not.(p(i) == 2 .and. x(i) < n - 1))
        i = i + 1
      end do
      if (i > n) call halt()
      a(i, j) = ibset(a(i, j), k)
      x(i) = x(i) + 1
      y(j) = y(j) + 1
    end do
  end do
  do i = 1, n
    write(*,'(i0)',advance='no') a(i, 1)
    do j = 2, n
      write(*,'(x,i0)',advance='no') a(i, j)
    end do
    write(*,*)
  end do
contains
  integer function pattern(s, b) result(res)
    integer, intent(in) :: s
    logical, intent(in) :: b
    ! 0: all 0
    ! 1: all 1
    ! 2: any 0
    ! 3: any 1
    res = mod(2 * s + merge(0, 1, b) + 1, 4)
  end
  subroutine halt()
    write(*,'(i0)') -1
    stop
  end
end program i_hate_matrix_construction